

ZdGravatar URL pour Real Studio

Mettez un Gravatar dans votre application

GravatarURL est une classe pour l'environnement de développement Real Studio et qui permet d'utiliser les services du site [Gravatar.com](https://gravatar.com) en générant automatiquement l'URL nécessaire pour récupérer l'avatar d'un membre simplement à partir de son adresse email.

La classe zdGravatarURL inclut la gestion de toutes les fonctionnalités mise à la disposition du programmeur par l'API de Gravatar pour la récupération des avatars:

- Grâce à l'utilisation d'une empreinte MD5, l'adresse email ne circule pas en clair sur le réseau
- Personnalisez la taille de l'avatar en fonction de vos besoins (1 à 2048 pixels).
- Utilisez une image par défaut, des avatars prédéfinis ou bien une image personnalisée si l'adresse est inconnue de Gravatar.
- Possibilité de forcer l'affichage de l'avatar par défaut à des fins de test
- Gestion du contrôle parental sur 4 niveaux: G, PG, R, X
- URL sécurisé SSL (https://) si besoin.
- Rajout optionnel d'une extension ".jpg" à l'URL à des fins de compatibilité.
- Aucune clé chiffrée (API key) ni aucune inscription ne sont nécessaires.

Guide d'utilisation de Gravatar URL

Comment utiliser zdGravatarURL ?

zdGravatarURL s'utilise de manière très simple. Il suffit d'importer le fichier de la classe zdGravatarURL dans votre projet Real Studio avec la commande *Importer* du menu *Fichier* ou bien en faisant glisser la classe zdGravatarURL d'un projet à l'autre.

Propriétés

• EmailAddress As String

Chaque utilisateur de Gravatar est identifié par son adresse email. Cette propriété doit donc contenir l'adresse email de l'utilisateur dont vous souhaitez obtenir l'avatar.

```
// Utilisation simple de zdGravatarURL
Dim theAvatar As New zdGravatarURL
theAvatar.EmailAddress = "exemple1@mondomaine.com"
HTMLViewer1.LoadPage theAvatar.AvatarURL
```

Important: L'adresse email n'apparaît jamais en clair dans l'URL générée par zdGravatarURL. Une empreinte (un "hash" MD5 sur 128 bits) est calculée à partir de cette adresse, et c'est cette empreinte, non réversible, qui est utilisée par les serveurs de Gravatar pour identifier le compte utilisateur.

• DefaultImage As String

Si l'empreinte de l'adresse email utilisée n'est pas reconnue par Gravatar, alors le serveur renvoie une image par défaut. Gravatar peut aussi renvoyer une image personnalisée de votre choix. Pour bénéficier des images prédéfinies par Gravatar, utilisez les constantes décrites dans le tableau ci-après.







```
// Utilisation de DefaultImage
Dim theAvatar As New zdGravatarURL
theAvatar.EmailAddress = "exemple1@mondomaine.com"
// L'image par défaut souhaitée est l'homme mystère
theAvatar.DefaultImage = zdGravatarURL.kDefaultMysteryMan
// Affiche l'avatar
HTMLViewer1.LoadPage theAvatar.AvatarURL
```

Vous avez aussi la possibilité d'utiliser une image par défaut personnalisée par vos soins. Par exemple une icône dont le style correspond à celui de votre application ou de votre site web. Pour cela il suffit tout simplement d'affecter votre URL à la propriété **DefaultImage**. Cette URL doit respecter quelques conditions simples:

1. Elle doit être accessible publiquement et sans procédure d'authentification HTTP. Elle ne peut donc pas être localisée sur un Intranet, une machine de développement, ou derrière un firewall.
2. Utiliser le protocole HTTP ou HTTPS sur le port standard. Respectivement 80 et 443.
3. Elle doit avoir une extension connue: .jpg, .jpeg, .gif, .png.
4. Si l'URL contient une requête GET, alors cette requête sera ignorée.

Afin d'éviter tout contenu malicieux, les images personnalisées subissent un scan de sécurité avant d'être renvoyé par le serveur de Gravatar.

```
// Utilisation de DefaultImage avec une URL d'image personnalisée
Dim theAvatar As New zdGravatarURL
theAvatar.EmailAddress = "exemple1@mondomaine.com"
// L'image par défaut souhaitée est définie par son URL
theAvatar.DefaultImage = "http://www.monsite.com/images/avatar.png"
// Affiche l'avatar
HTMLViewer1.LoadPage theAvatar.AvatarURL
```

Avatar	Constante	Valeur	Description
	kDefaultGravatar	"" (Une chaîne vide)	C'est l'image générique renvoyé par Gravatar si vous n'utilisez pas la propriété DefaultImage de zdGravatarURL.
	kDefaultMysteryMan	"mm"	Silhouette appelée "Mystery-Man". Cette image est fixe et ne dépend pas de l'empreinte de l'adresse email.
	kDefaultIdentIcon	"identicon"	Une figure géométrique utilisant des fractales calculées en fonction de l'empreinte de l'adresse email.
	kDefaultMonsterID	"monsterid"	Une image de «monstre» dont la forme et les couleurs sont basées sur l'empreinte de l'adresse email.
	kDefaultWavatar	"wavatar"	Des figures différentes générées en fonction de l'empreinte de l'adresse email.
	kDefaultRetro	"retro"	Une icône 8-bits avec un look rétro, générée en fonction de l'empreinte de l'adresse email.
	kDefaultBlank	"blank"	Renvoie une image transparente de type PNG. Cette image est donc invisible, mais aux bonnes dimensions.
Aucun	kDefault404	"404"	Renvoie simplement une erreur HTTP 404.

• ForceDefault As Boolean

Si cette propriété est **True**, le serveur de Gravatar renverra toujours l'image par défaut telle que définie par la propriété **DefaultImage**.

```
// Utilisation de ForceDefault
Dim theAvatar As New zdGravatarURL
theAvatar.EmailAddress = "exemple1@mondomaine.com"
theAvatar.DefaultImage = zdGravatarURL.kDefaultWavatar

#if DebugBuild Then
    // Force l'affichage de l'image par défaut si l'exécution a lieu dans l'IDE.
    theAvatar.ForceDefault = True
#endif

HTMLViewer1.LoadPage theAvatar.AvatarURL
```

• Rating As String

Gravatar demande à ses utilisateurs de spécifier un niveau de contrôle parental pour les icônes personnelles. En utilisant zdGravatarURL, vous pouvez faire bénéficier votre application de ce contrôle parental en spécifiant une limite au type de contenu que vous souhaitez afficher.

Gravatar utilise quatre niveaux de contrôle parental correspondant à quatre constantes définies par la classe `zdGravatarURL` et décrites dans le tableau ci-après.

Le niveau par défaut est "G", c'est à dire "tout public". Si vous spécifiez un autre niveau, Gravatar retournera une image correspondant, soit à ce niveau, soit à un niveau inférieur. Si l'utilisateur de Gravatar identifié par l'empreinte de son adresse email ne possède pas d'image correspondante ou de niveau inférieur, alors Gravatar renvoie l'image par défaut.

Par exemple, si vous réglez la propriété `Rating` avec la constante `kRatingPG`, alors Gravatar vous renverra soit une image définie comme de niveau PG ou G. Si le compte ne dispose que d'images R ou X, alors Gravatar enverra l'image par défaut.

Niveau	Constante	Valeur	Description	Peut renvoyer
G	<code>kRatingG</code>	"g"	Convient à tous les publics.	G
PG	<code>kRatingPG</code>	"pg"	Gestes grossiers, jurons, tenues provocantes, violence.	G, PG
R	<code>kRatingR</code>	"r"	Blasphème, grande violence, nudité, usage de stupéfiants.	G, PG, R
X	<code>kRatingX</code>	"x"	Très grande violence, pornographie.	G, PG, R, X

• Size As Integer

Par défaut, l'image retournée par Gravatar est à la taille de 80 x 80 pixels. Si vous souhaitez obtenir une taille différente (de 1 à 2048 pixels), vous devez affecter la taille désirée à la propriété **Size**. Pour revenir à la valeur par défaut, utilisez la constante **kSizeDefault**.

```
// Utilisation de la propriété Size
Dim theAvatar As New zdGravatarURL
theAvatar.EmailAddress = "exemple1@mondomaine.com"
// We need a 256 x 256 pixel image
theAvatar.Size = 256
Dim theURLString As String = theAvatar.AvatarURL
```

• UseSSL As Boolean

Le serveur de Gravatar peut aussi fournir les images par le biais d'un serveur HTTP sécurisé par SSL. Pour bénéficier de cette possibilité, affectez la valeur **True** à cette propriété afin d'obtenir une URL de la forme "https://".

Méthode partagée (Shared Method)

• Function VersionString() As String (à partir de v1.1)

Retourne la version de `zdGravatarURL` sous forme de chaîne. La version 1.1 renvoie la chaîne "1.1".

Méthodes

• Function EmailAddressHash(inEmailAddress As String = "") As String (à partir de v1.1)

Retourne l'empreinte MD5 de l'adresse email passée en paramètre. Si cette adresse est vide, ou absente, alors cette fonction retourne l'empreinte MD5 de la chaîne contenue dans la propriété **EmailAddress**.

• Function URL() As String

Retourne l'URL de l'avatar en fonction de l'ensemble des options. Après avoir créé une instance de la classe et ajusté les différents paramètres, vous devez appeler cette méthode afin de générer l'URL qui vous permettra de récupérer l'avatar désiré.

```
// Utilisation simple de zdGravatarURL
Dim theAvatar As New zdGravatarURL
theAvatar.EmailAddress = "exemple1@mondomaine.com"
HTMLViewer1.LoadPage theAvatar.AvatarURL
```

- **Function URL(inEmailAddress As String) As String**

Cette fonction permet de générer l'URL de l'avatar correspondant à l'adresse email passée en paramètre, **sans changer le contenu de la propriété EmailAddress.**

L'exemple ci-après décrit l'utilisation de `zdGravatarURL` pour générer un tableau d'URLs, avec le même réglage d'options, à partir d'un tableau d'adresse email.

```
Function BuildURLsArray(inAddresses() As String) As String()

    '-- Return an array of Gravatar URLs generated from an array of addresses

    // Instantiate and setup a zdGravatarURL
    Dim theAvatar As New zdGravatarURL
    theAvatar.DefaultImage = zdGravatarURL.kDefaultIconMonsterID
    theAvatar.Rating = zdGravatarURL.kRatingPG
    theAvatar.Size = 128

    // Creating an array of URLs of the same size
    Dim theLastIndex As Integer = inAddresses.Ubound
    Dim theURLs() As String
    Redim theURLs( theLastIndex )

    // Loop through all provided addresses
    For i As Integer = 0 To theLastIndex
        // Generate the i-th URL
        theURLs( i ) = theAvatar.URL( inAddresses( i ) )
    Next

    // Return the result
    Return theURLs

End Function
```

Configuration requise

`zdGravatarURL` fonctionne avec tous les types de projets disponibles et connus à ce jour dans Real Studio (console, application graphique, application web, etc...) à partir de la version 2009r4 et sur toutes les plateformes disponibles: MacOS, Windows, Linux

License:

`zdGravatar` est publié sous la licence X11. Consultez le fichier `LICENSE.txt` pour plus d'informations (en Anglais).

Contact:

Le projet `zdGravatarURL` est hébergé à l'adresse <http://code.google.com/p/zd-gravatarurl/>. pour toute question ayant trait à `zdGravatarURL`, envoyez un mail à l'adresse commençant par 'zaatar' et affichée dans la liste des membres sur la gauche de la page d'accueil du projet. Si vous trouvez un bug dans `zdGravatarURL` ou une erreur dans cette documentation, merci de bien vouloir la signaler à l'adresse suivante : <http://code.google.com/p/zd-gravatarurl/issues/list>.